Aristotle's Assertoric Syllogistic

Angeliki Koutsoukou-Argyraki

December 14, 2021

Abstract

We formalise with Isabelle/HOL some basic elements of Aristotle's assertoric syllogistic following the article from the Stanford Encyclopedia of Philosophy by Robin Smith: https://plato.stanford.edu/entries/aristotle-logic/. To this end, we use a set theoretic formulation (covering both individual and general predication). In particular, we formalise the deductions in the Figures and after that we present Aristotle's metatheoretical observation that all deductions in the Figures can in fact be reduced to either Barbara or Celarent. As the formal proofs prove to be straightforward, the interest of this entry lies in illustrating the functionality of Isabelle and high efficiency of Sledgehammer for simple exercises in philosophy.

Contents

1 Aristotle's Assertoric Syllogistic

theory *AristotlesAssertoric*
 imports *Main*
begin

1.1 Aristotelean Categorical Sentences

Aristotle's universal, particular and indefinite predications (affirmations and denials) are expressed here using a set theoretic formulation. Aristotle handles in the same way individual and general predications i.e. he gives the same logical analysis to "Socrates is an animal" and "humans are animals". Here we define the general predication i.e. predications are defined as relations between sets. This has the benefit that individual predication can also be expressed as set membership (e.g. see the lemma SocratesMortal).

definition *universal-affirmation* :: $'a\ set \Rightarrow 'a\ set \Rightarrow bool$ (**infixr** *Q 80*)
 where $A\ Q\ B \equiv \forall\ b \in B\ .\ b \in A$

definition *universal-denial* :: $'a\ set \Rightarrow 'a\ set \Rightarrow bool$ (**infixr** *E 80*)
 where $A\ E\ B \equiv \forall\ b \in B.\ (\ b \notin A)$

definition *particular-affirmation* :: $'a\ set \Rightarrow 'a\ set \Rightarrow bool$ (**infixr** *I 80*)
 where $A\ I\ B \equiv \exists\ b \in B.\ (\ b \in A)$

definition *particular-denial* :: $'a\ set \Rightarrow 'a\ set \Rightarrow bool$ (**infixr** *Z 80*)
 where $A\ Z\ B \equiv \exists\ b \in B.\ (\ b \notin A)$

The above four definitions are known as the "square of opposition".

definition *indefinite-affirmation* :: $'a\ set \Rightarrow 'a\ set \Rightarrow bool$ (**infixr** *QI 80*)
 where $A\ QI\ B \equiv ((\ \forall\ b \in B.\ (b \in A)) \lor\ (\exists\ b \in B.\ (b \in A)))$

definition *indefinite-denial* :: $'a\ set \Rightarrow 'a\ set \Rightarrow bool$ (**infixr** *EZ 80*)
 where $A\ EZ\ \ B \equiv ((\ \forall\ b \in B.\ (b \notin A)) \lor (\exists\ b \in B.\ (b \notin A)))$

lemma *aristo-conversion1* :
 assumes $A\ E\ B$ **shows** $B\ E\ A$
 using *assms universal-denial-def* **by** *blast*

lemma *aristo-conversion2* :
 assumes $A\ I\ B$ **shows** $B\ I\ A$
 using *assms* **unfolding** *particular-affirmation-def*
 by *blast*

lemma *aristo-conversion3* : **assumes** $A\ Q\ B$ **and** $B \neq \{\}$ **shows** $B\ I\ A$
 using *assms*
 unfolding *universal-affirmation-def particular-affirmation-def* **by** *blast*

Remark: Aristotle in general supposes that sets have to be nonempty. Indeed, we observe that in many instances it is necessary to assume that the sets are nonempty, otherwise Isabelle's automation finds counterexamples.

1.2 The Deductions in the Figures ("Moods")

The medieval mnemonic names are used.

1.2.1 First Figure

lemma *Barbara*:
assumes $A\ Q\ B$ **and** $B\ Q\ C$ **shows** $A\ Q\ C$
by (*meson assms universal-affirmation-def*)

lemma *Celarent*:
assumes $A\ E\ B$ **and** $B\ Q\ C$ **shows** $A\ E\ C$
by (*meson assms universal-affirmation-def universal-denial-def*)

lemma *Darii*:
assumes $A\ Q\ B$ **and** $B\ I\ C$ **shows** $A\ I\ C$
by (*meson assms particular-affirmation-def universal-affirmation-def*)

lemma *Ferio*:
assumes $A\ E\ B$ **and** $B\ I\ C$ **shows** $A\ Z\ C$
by (*meson assms particular-affirmation-def particular-denial-def universal-denial-def*)

1.2.2 Second Figure

lemma *Cesare*:
assumes $A\ E\ B$ **and** $A\ Q\ C$ **shows** $B\ E\ C$
using *Celarent aristo-conversion1 assms* **by** *blast*

lemma *Camestres*:
assumes $A\ Q\ B$ **and** $A\ E\ C$ **shows** $B\ E\ C$
using *Cesare aristo-conversion1 assms* **by** *blast*

lemma *Festino*:
assumes $A\ E\ B$ **and** $A\ I\ C$ **shows** $B\ Z\ C$
using *Ferio aristo-conversion1 assms* **by** *blast*

lemma *Baroco*:
assumes $A\ Q\ B$ **and** $A\ Z\ C$ **shows** $B\ Z\ C$
by (*meson assms particular-denial-def universal-affirmation-def*)

1.2.3 Third Figure

lemma *Darapti*:
assumes $A\ Q\ C$ **and** $B\ Q\ C$ **and** $C \neq \{\}$ **shows** $A\ I\ B$
using *Darii assms* **unfolding** *universal-affirmation-def particular-affirmation-def*
by *blast*

lemma *Felapton*:
assumes $A\ E\ C$ **and** $B\ Q\ C$ **and** $C \neq \{\}$ **shows** $A\ Z\ B$
using *Festino aristo-conversion1 aristo-conversion3 assms* **by** *blast*

lemma *Disamis*:
assumes $A\ I\ C$ **and** $B\ Q\ C$ **shows** $A\ I\ B$
using *Darii aristo-conversion2 assms* **by** *blast*

lemma *Datisi*:
assumes $A\ Q\ C$ **and** $B\ I\ C$ **shows** $A\ I\ B$
using *Disamis aristo-conversion2 assms* **by** *blast*

lemma *Bocardo*:
assumes $A\ Z\ C$ **and** $B\ Q\ C$ **shows** $A\ Z\ B$
by (*meson assms particular-denial-def universal-affirmation-def*)

lemma *Ferison*:
assumes $A\ E\ C$ **and** $B\ I\ C$ **shows** $A\ Z\ B$
using *Ferio aristo-conversion2 assms* **by** *blast*

1.2.4 Examples

Example of a deduction with general predication.

lemma *GreekMortal* :
assumes *Mortal Q Human* **and** *Human Q Greek*
shows *Mortal Q Greek*
using *assms Barbara* **by** *auto*

Example of a deduction with individual predication.

lemma *SocratesMortal*:
assumes *Socrates* $\in$ *Human* **and** *Mortal Q Human*
shows *Socrates* $\in$ *Mortal*
using *assms* **by** (*simp add*: *universal-affirmation-def*)

1.3 Metatheoretical comments

The following are presented to demonstrate one of Aristotle's metatheoretical explorations. Namely, Aristotle's metatheorem that: "All deductions in all three Figures can eventually be reduced to either Barbara or Celarent" is demonstrated by the proofs below and by considering the proofs from the previous subsection.

lemma *Darii-reducedto-Camestres*:
assumes $A\ Q\ B$ **and** $B\ I\ C$ **and** $A\ E\ C$
shows $A\ I\ C$
proof−
have $B\ E\ C$ **using** *Camestres* ‹ $A\ Q\ B$ › ‹$A\ E\ C$› **by** *blast*
show *?thesis* **using** ‹ $B\ I\ C$ › ‹$B\ E\ C$›
by (*simp add*: *particular-affirmation-def universal-denial-def*)
qed

It is already evident from the proofs in the previous subsection that:
Camestres can be reduced to Cesare.
Cesare can be reduced to Celarent.
Festino can be reduced to Ferio.

lemma *Ferio-reducedto-Cesare*: **assumes**

```
  A E B  and B I C and A Q C
shows A Z C
 proof−
  have B E C using Cesare ‹A E B › ‹A Q C› by blast
  show ?thesis using ‹B I C › ‹B E C›
    by (simp add: particular-affirmation-def universal-denial-def)
qed

lemma Baroco-reducedto-Barbara :
  assumes A Q B and A Z C and B Q C
  shows B Z C
proof−
  have A Q C using ‹A Q B › ‹ B Q C › Barbara by blast
  show ?thesis using ‹A Q C› ‹ A Z C ›
    by (simp add: particular-denial-def universal-affirmation-def)
qed

lemma Bocardo-reducedto-Barbara :
  assumes A Z C and B Q C and A Q B
  shows A Z B
proof−
  have A Q C using ‹B Q C› ‹ A Q B› using Barbara by blast
  show ?thesis using ‹A Q C› ‹ A Z C›
    by (simp add: particular-denial-def universal-affirmation-def)
qed
```

Finally, it is already evident from the proofs in the previous subsection that :

Darapti can be reduced to Darii.
Felapton can be reduced to Festino.
Disamis can be reduced to Darii.
Datisi can be reduced to Disamis.
Ferison can be reduced to Ferio.

In conclusion, the aforementioned deductions have thus been shown to be reduced to either Barbara or Celarent as follows:

Baroco ⇒ Barbara
Bocardo ⇒ Barbara
Felapton ⇒ Festino ⇒ Ferio ⇒ Cesare ⇒ Celarent
Datisi ⇒ Disamis ⇒ Darii ⇒ Camestres ⇒ Cesare
Darapti ⇒ Darii
Ferison ⇒ Ferio

1.4 Acknowledgements

A.K.-A. was supported by the ERC Advanced Grant ALEXANDRIA (Project 742178) funded by the European Research Council and led by Professor Lawrence Paulson at the University of Cambridge, UK. Thanks to Wenda

Li.

1.5 Bibliography

Smith, Robin, "Aristotle's Logic", The Stanford Encyclopedia of Philosophy (Summer 2019 Edition), Edward N. Zalta (ed.), URL = https://plato.stanford.edu/archives/sum2019/entries/aristotle-logic/

end